Using SAS Software to Determine the Mixed Rotterdam Demand Model Parameters and Test Statistical References

Maryam Tabarestani, Louisiana State University[1]
Hector O. Zapata, Louisiana State University

Abstract

Although SAS as strong statistical software can be used to model mixed demand model, so far all studies have applied TSP and other software to analyze these types of models. This paper demonstrates for the first time the use of SAS to estimate mixed demand models. A mixed demand system (MDS) is one type of demand functions used in applied demand analysis, the model which was introduced by Samuelson in 1965. A MDS is a more flexible functional form for many raw and processed agricultural commodities. Samuelson's mixed demand model also provides a theoretical basis for applied demand studies that use time series data. Although a MDS is a more generalized functional form for modeling consumer demand, very little work has been published on the estimation of such models with popular computer software such as SAS. But there is a large body of literature based on the use of other similar models. For example, the Rotterdam parameterization is one of the differential approaches introduced by Theil (1965) and Barten (1969). However, it may not well known that the Rotterdam demand model is able to satisfy all mixed demand assumptions and restrictions. One reason for the paucity in using a mixed Rotterdam demand specification is the degree of mathematical and computational requirements. This paper tries to bridge the gap between theory and practical applications of MDS by illustrating the adaption of SAS statements to calculate the mixed Rotterdam demand variables (the transformed variables that are needed for final model estimation), the commands required to estimate the equations of the system, and an illustration of testing restrictions and model simulation.

Introduction

The literature on applied demand analysis and the estimation of demand systems has been briefly presented in Deaton and Meullbahoer (1980) and Theil and Clements (1987). In these

---

[1] Maryam Tabarestani obtained her Ph.D. from the Department of Agricultural Economics & Agribusiness, and Hector O. Zapata is the Executive Director of International Programs and Past Presidents of the LSU Alumni Association Alumni Professor, Louisiana State University, Baton Rouge, Louisiana.

books and in other published papers, the typical form of one equation is to model per-capita consumption (or some other left hand side variable) as a function of prices, income, and other demographic variables. The bulk of such work builds systems of equations that are direct as in the above equation (Marshallian demand) or inverse (price as dependent variable). A third kind of demand models that combines these two functional forms is known as MDS (Mocshini and Rizzi, 2007) and have been proposed as alternative more flexible structures.

The mixed demand concept in consumer behavioral theory was first introduced by Samuelson (1965). This specification, which regards demand relationships as a function of mixed groups of quantities and prices, provides a more flexible specification than estimating a simultaneous system of demand and supply equations (Moschini and Vissa, 1993). A set of mixed demand equations contains the coefficients of a regular demand and of an inverse demand system together (Barton, 1992).

MDS have been used effectively in market structure studies of several agricultural commodities such as meat/livestock products (Heien, 1977; Moschini and Vissa, 1993; McLaren and Gary Wong, 2009), fruits (Brown and Lee, 2006), vegetables (Moschini and Rizzi, 2007; Barten, 1992), transportation (Cunha-e-Sa and Ducla-Soares, 1999; Cunha-e-Sa et al., 2004), and rationed and non-rationed food (Gao, et al., 1996). All of these research papers have shown the primary importance of specifying a consistent model in empirical demand studies. In the case of estimating a MDS, there is the additional requirement of specifying a mixed utility function.

A mixed Rotterdam demand specification needs a broad knowledge of mathematics, statistics, and an appropriate statistical software package such as SAS. Mocshini and Vissa (1993), Moschini and Rissi (2006, 2007), and Gary Wong and Park (2007) have obtained an empirical model for the mixed demand through a conditional cost function. Mixed Rotterdam Demand System (MRDS), Stone-Geary Mixed Demand System, Normalized-Quadratic Mixed Demand System, Quadratic Almost Ideal Mixed Demand System (QAIMDS), and Nested Constant Elasticity of Substitution (NCES) are some examples of models that have been derived from this technique. Brown et al. (1993), Brown and Lee (2006), and Gao et al. (1996) parameterize a MRDS via the primary utility functional form, as Samuelson (1965) and Chavas (1984) analysis, and statistical software such as TSP and SHAZAM. This article first provides some explanations about the economic theory behind the MRDS and the econometric techniques for model estimation. This is followed with the SAS programming needed for estimating a MRDS, calculating model prerequisite variables, estimating model parameters and elasticities, and testing the reliability of the estimates.

Methodology

Mixed Demand Specification

The mixed demand system of equations, like ordinary demand functions, can also be derived from the consumer utility maximization problem. The mixed set of demand functions contains both coefficients of a Marshallian demand system and of an inverse demand system (Barton, 1992). In the case of deriving a mixed demand system, one needs to specify a mixed utility function also which can be written as follows:

$$\underset{q_a, p_b}{Max} \qquad U(q_a, q_b, p_a, p_b, y) = u(q_a, q_b) - v(p_a, p_b, M) \tag{1a}$$

$$s.t. \qquad p'_a q_a + p'_b q_b = M$$

where the functions, u(.) direct utility, and, v(.) indirect utility, are dual functions. Therefore, the function "U" is a function of either the quantity set or the relative price set. To find the set of all n equilibrium points, one should maximize a total space of 2n-dimension of (p_{ai}, q_{bj}), with respect to $Q'P = M$, where i=1, 2, …, n and j=1, 2, .., m (Samuelson, 1965).

$$L = u(q_a, q_b) - v(p_a, p_b, M) - \lambda(M - p'_a q_a - p'_b q_b), \tag{2a}$$

$$\frac{\partial z(q_a^*, q_b, p_a, p_b^*)}{\partial q_a} = \frac{\partial u(q_a^* . q_b)}{\partial q_a} - \lambda p_a = 0 \tag{2b}$$

$$\frac{\partial z(q_a^*, q_b, p_a, p_b^*)}{\partial p_b} = -\frac{\partial v(p_b, p_b^*)}{\partial p_b} - \lambda q_b = 0 \tag{2c}$$

$$\frac{\partial z(q_a^*, q_b, p_a, p_b^*)}{\partial \lambda} = p'_a q_a + p'_b q_b - M = 0 \tag{2d}$$

where λ is the Lagrangian multiplier. This system includes "n+m+1" equations and unknown parameters. Conditioning on the budget constraint, these "n+m+1" equations, obtained from first derivative of optimization/maximization of Lagrangian expression (equation 2a), insure that the utility function is stationary. The second order condition, which is the second derivative of the optimization problem, guarantees that the stationary value is a maximum (Theil, 1975, 1976). After solving three equations, 2b, 2c, 2d, simultaneously, the result will be a mixed demand system.

$$q_a^* = q_a(\bar{p}_a, \bar{q}_b, M) \tag{3a}$$

$$p_b^* = p_b(\bar{p}_a, \bar{q}_b, M) \tag{3b}$$

$$\lambda = \lambda(\bar{p}_a, \bar{q}_b, M). \tag{3c}$$

Mixed Rotterdam Demand Specification under Preference Independence Assumption

To apply the mixed demand system for the practical determination with observed variables, one needs a dual flexible representation of preferences, since the solutions of optimality conditions (equations 2) are not feasible as general representations of preferences (Mocshini and Rizzi, 2007).

To obtain a mixed Rotterdam demand system, the following steps are required. First a total differentiation of the mixed demands (3a) and (3b) is derived:

$$dq_a = (\frac{\partial q_a}{\partial p'_a})dp_a + (\frac{\partial q_a}{\partial q'_b})dq_b \tag{4a}$$

$$dp_b = (\frac{\partial p_b}{\partial p'_a})dp_a + (\frac{\partial p_b}{\partial q'_b})dq_b \tag{4b}$$

$$d\lambda = (\frac{\partial \lambda}{\partial p'_a})dp_a + (\frac{\partial \lambda}{\partial q'_b})dq_b.$$

In the second step, the total differentiations of the first order conditions, equations 4, must be derived and are given by:

$$\frac{\partial^2 u(q_a^*.\bar{q}_b)}{\partial q_a \partial q_a} dq_a + \frac{\partial^2 u(q_a^*.\bar{q}_b)}{\partial q_a \partial q_b} dq_b = \lambda dp_a + p_a d\lambda \tag{5a}$$

$$\frac{\partial^2 v(\bar{p}_b, p_b^*)}{\partial p_b \partial p_b} dp_b + \frac{\partial^2 v(\bar{p}_b, p_b^*)}{\partial p_b \partial p_a} dp_a = -\lambda dq_b - q_b d\lambda \tag{5b}$$

$$p'_a dq_a + q'_a dp_b + p'_b dq_b + q_b dp_b = dM.$$

Then, after rearranging terms and algebraic manipulations, the following differential mixed demand system with nominal prices is obtained:

$$w_{ai} Dq_{ai} = \theta_i DQ + \phi\theta_i (Dp_{aj} - \sum_j \theta_j Dp_{aj} + \sum_s \gamma_s Dq_{bs}) \tag{6a}$$

$$w_{br} Dp_{br} = -\gamma_r DQ - \phi\gamma_r (Dq_{br} - \sum_j \theta_j Dp_{aj} + \sum_s \gamma_s Dq_{bs}) \tag{6b}$$

These equations represent the Rotterdam specification of the mixed demand system. The parameters of such system are non-linear while variables are linear. These two functional forms are based on Preference Independence restriction for the direct and indirect utility functions,

The subscripts i and j stand for the elements (quantities or prices) of the commodity group "a" and subscripts r and s represent the elements (quantities or prices) of the commodity group "b." All differential equations as well as Rotterdam demand system should satisfy the classic restrictions of consumer behavior theory. The parametric restrictions of this system are given by:

- Engel aggregation/Adding up:

$$\sum_i \theta_i - \sum_r \gamma_r = 1 \qquad \sum_i \theta_{ij} = \theta_j \qquad \sum_r \gamma_{rs} = \gamma_s \tag{7}$$

- Homogeneity:

$$\sum_i \theta_{ij} = \theta_j \qquad \sum_r \gamma_{rs} = \gamma_s \tag{8}$$

- Symmetry:

$$\gamma_{rs} = \gamma_{sr} \qquad \theta_{ij} = \theta_{ji}. \tag{9}$$

Mixed Rotterdam Demand Parameters, Variables, and Elasticities

- Parameters

The term ϕ is the reciprocal of the income elasticity of the marginal utility of income (Theil, 1987). The terms $\theta_i = \frac{\partial(p_{aj}q_{ai})}{\partial M}$ and $\gamma_r = \frac{\partial(p_{br}q_{br})}{\partial M}$ refer to the marginal budget shares (Theil, 1987) for the commodity group "a" and "b."

- Variables

$$DQ = \sum_i w_{ai} Dq_{ai} + \sum_r w_{br} Dp_{br} \tag{10a}$$

$$\widehat{p}_a \widehat{M}^{-1} \widehat{q}_a = [\frac{p_{ai}q_{ai}}{M}] = [w_{ai}] \tag{10b}$$

$$\widehat{q}_b \widehat{M}^{-1} \widehat{p}_b = [\frac{p_{br}q_{br}}{M}] = [w_{br}] \tag{10c}$$

$$\hat{q}_a^{-1} dq_a = [\frac{dq_{ai}}{q_{ai}}] = [d\,log(q_{ai})] = [Dq_{ai}] \tag{10d}$$

$$\hat{q}_b^{-1} dq_b = [\frac{dq_{br}}{q_{br}}] = [d\,log(q_{br})] = [Dq_{br}] \tag{10e}$$

$$\hat{p}_b^{-1} dp_b = [\frac{dp_{br}}{p_{br}}] = [d\,log(p_{br})] = [Dp_{br}] \tag{10f}$$

$$\hat{p}_a^{-1} dp_a = [\frac{dp_{ai}}{p_{ai}}] = [d\,log(p_{ai})] = [Dp_{ai}]\,. \tag{10g}$$

- Elastisities

✓ own-price elasticities for group a commodities:

$$\varepsilon_{ii} = \frac{d\,log(q_{ai})}{d\,log\;p_{ai}} = \frac{Dq_{ai}}{Dp_{ai}} = \frac{\phi}{w_{ai}}(\theta_i - \theta_i * \theta_i) = \frac{\phi}{w_{ai}}(\theta_i - \theta_i * \theta_i) \tag{11a}$$

✓ cross-price elasticity for group a commodities:

$$\varepsilon_{ij} = \frac{d\,log(q_{ai})}{d\,log\;p_{aj}} = \frac{Dq_{ai}}{Dp_{aj}} = \frac{\phi}{w_{ai}}(-\theta_i * \theta_j) \tag{11b}$$

✓ quantity elasticity for group a commodities:

$$\delta_{ij} = \frac{d\,log(q_{ai})}{d\,log\;q_{bs}} = \frac{Dq_{ai}}{Dq_{bs}} = \frac{\phi}{w_{ai}}(\theta_i * \gamma_s) \tag{11c}$$

✓ price elasticity for commodity group b:

$$\varsigma_{ri} = \frac{d\,log(p_{br})}{d\,log\;p_{ai}} = \frac{Dp_{br}}{Dp_{ai}} = \frac{\phi}{w_{br}}(\gamma_r * \theta_i) \tag{11d}$$

✓ own-quantity elasticity/price flexibility for commodity group b:

$$\varsigma_{rr} = \frac{d\,log(p_{br})}{d\,log\;q_{br}} = \frac{Dp_{br}}{Dq_{br}} = -\frac{\phi}{w_{br}}(\gamma_r + \gamma_r * \gamma_r) \tag{11e}$$

✓ cross-quantity elasticity/price flexibility for group b commodities:

$$\varsigma_{rs} = \frac{d\,log(p_{br})}{d\,log\;q_{bs}} = \frac{Dp_{br}}{Dq_{bs}} = -\frac{\phi}{w_{br}}(\gamma_r * \gamma_s) \tag{11f}$$

✓ Income elasticity for both groups a and b:

$$\eta_{ai} = \frac{\theta_i}{w_{ai}} \tag{11g}$$

$$\eta_{br} = \frac{-\gamma_r}{w_{br}}. \tag{11h}$$

Mixed Rotterdam Demand Econometrics Model Specification

Theil (1987) posits that the Rotterdam demand system is a finite-change version of a differential demand system which is in terms of infinitesimal changes; $Dx_t = \log x_t - \log x_{t-1}$; t=time, for any positive variable x, $\overline{w}_{ait} = \frac{w_{ai,t-1} + w_{ait}}{2}$, $\overline{w}_{brt} = \frac{w_{br,t-1} + w_{brt}}{2}$ and $DQ_t = \sum_i \overline{w}_{ait} Dq_{ait} + \sum_{r=1}^{m} \overline{w}_{brt} Dp_{brt},$. Therefore, the equations below are empirical applications, in the form of econometric specification, of the mixed Rotterdam demand model in this study:

$$\overline{w}_{ait} Dq_{ait} = \theta_i DQ_t + \phi\theta_i (Dp_{ajt} - \sum_j \theta_j Dp_{ajt} + \sum_s \gamma_s Dq_{bst}) + \varepsilon_{ait} \tag{12a}$$

$$\overline{w}_{brt} Dp_{brt} = -\gamma_r DQ_t - \phi\gamma_r (Dq_{brt} - \sum_j \theta_j Dp_{ajt} + \sum_s \gamma_s Dq_{bst}) + \varepsilon_{brt} \tag{12b}$$

where the subscripts t imply to the Theil's definitions. That is, a discrete change from period t-1 to t. The terms $\overline{w}_{ait}$ and $\overline{w}_{brt}$ are the arithmetic average of $w_{ai,t-1}$ and $w_{ai,t}$ and $w_{br,t-1}$ and $w_{br,t}$, respectively. The disturbance terms, ε_{ait} and ε_{brt} are represented as the stochastic process of the econometric model. All other variables and parameters are as defined in previous sections.

Application of the Mixed Rotterdam Demand System and Data

The mixed Rotterdam demand system is used to analyze the determinants of the U.S. shrimp industry, after providing its econometrics framework. In general, two different shrimp products are supplied in the U.S. market: domestically produced shrimp and imported shrimp.

While a mixed Rotterdam demand contains both Marshallian/regular demand and inverse demand equations, the Gulf of Mexico shrimp production is modeled in quantity-predetermined framework (inverse demand equations) and the U.S. imports has been modeled in a price-predetermined framework (Marshallian demand). The mixed Rotterdam demand system is estimated using time series data for U.S. shrimp imports and Gulf of Mexico shrimp landings quantities and prices. The data for U.S. imports consist of eight countries including Thailand, Vietnam, China, India, Indonesia, Ecuador, Mexico, and a final category includes all other exporter countries. Demand for Gulf shrimp is specified by size of shrimp with three sizes (large, medium, and small) considered. The data are quarterly for 1995(1) and ends for 2010(1).

SAS Programming

This section provides a complete description of the SAS programming used in the estimation of the mixed Rotterdam demand system. Sufficient details are provided so that practitioners can adapt the program to their needs without technical assistance.

Data Description

A descriptive statistical analysis is important to gain familiarity with the time series patterns for variables in the model and for a better insight into the model structure. The following SAS statements are standard procedures in other SAS applications.

```
proc import datafile = 'C:\Users\maryam\Desktop\Landings New Data_BPW-
1980-2012-SASfile-Nov29.xls'
OUT=Mix DBMS=excel2000 replace;
run;
proc means data=mix1 mean;
      var   Chi_PPP    Chi_PO     Ecu_PPP     Ecu_PO
            Indi_PPP   Indi_PO    Indo_PPP    Indo_PO
            Mex_PPP    Mex_PO     Thi_PPP     Thi_PO
            Vie_PPP    Vie_PO     Other_PPP   Other_PO

            L1_PPP     L1_PO
            L2_PPP     L2_PO
            L3_PPP     L3_PO;
run;
```

Transforming Variables for the Mixed Rotterdam Demand Model

The units for the original prices in the dataset were in total values. To following SAS commands transform prices to a per pound basis:

```
/**********************************************************************
*                        Price per Pound calculation                 *
**********************************************************************/

data mix2; set mix1;
      Chi_PPP  = Chi_PR/Chi_PO;
      Ecu_PPP  = Ecu_PR/Ecu_PO;
      .
      .
      .
      L3_PPP   = L3_PR/L3_PO;
run;
```

To calculate the share of each type of shrimp product in total expenditures, first total expenditures should be calculated. The SAS statements for calculation of total expenditures, each shrimp product's budget share, and their mean values are provided as follows:

```
/**********************************************************************
*                  Total Expenditure/Income Calculation              *
**********************************************************************/

data mix3; set mix2;
      M = Chi_PR + Ecu_PR + Indi_PR + Indo_PR + Mex_PR + Thi_PR
        + Vie_PR + Other_PR + L1_PR + L2_PR + L3_PR;
run;

/**********************************************************************
*                        Budget Shares Calculation                   *
```

$$* \qquad \hat{p}_a \hat{M}^{-1} \hat{q}_a = [\frac{p_{ai} q_{ai}}{M}] = [w_{ai}] \qquad \hat{q}_b \hat{M}^{-1} \hat{p}_b = [\frac{p_{br} q_{br}}{M}] = [w_{br}] \qquad *$$

```
**********************************************************************/

data mix4; set mix3;
      W1   = (Chi_PPP * Chi_PO)/M;
      W2   = (Ecu_PPP * Ecu_PO)/M;
```

```
    .
    .
    .
    W11 = (L3_PPP * L3_PO)/M;

    LW1=lag(W1);
    LW2=lag(W2);
    .
    .
    .
    LW11=lag(W11);

    Wbar1  = (W1 + LW1) /2;
    Wbar2  = (W2 + LW2) /2;
    .
    .
    .
    Wbar11 = (W11 + LW11)/2;
run;

proc means data=mix4 mean;
    title 'Average Data for Estimation of Elasticities';
    var  wbar1 Wbar2 … wbar10 wbar11;
    output out = meandata
    mean = mw1 mw2 ... mw11
           wbar1 Wbar2 ... wbar11;
run;
```

Another step in implementing the mixed Rotterdam demand system is the conversion of raw data to logarithmic differences. In this paper quarterly data are used, so log differences are quarterly first differences. An appropriate SAS code for this computation is demonstrated below:

```
data mix4; set mix3;
/*********************************************************************
*                  Independent Variables Calculation                  *
*                  [d log( p_ai )] = [Dp_ai ]      [d log( q_br )] = [Dq_br ]              *
*********************************************************************/

     LC_PPP  = log(Chi_PPP);           dC_PPP  = dif(LC_PPP);
     LEc_PPP = log(Ecu_PPP);           dEc_PPP = dif(LEc_PPP);
     .                                 .
     .                                 .
     .                                 .
     LO_PPP  = log(Other_PPP);         dO_PPP  = dif(LO_PPP);
     .                                 .
     .                                 .
     LL3_PO  = log(L3_PO);             dL3_PO  = dif(LL3_PO);

/**********************************************************************
*                   Dependent Variables Calculation                    *
*  [w_ai ]*[d log(q_ai )] = [w_ai ]*[Dq_ai ]    [w_br ]*[d log( p_br )] = [w_br ]*[Dp_br ]   *
**********************************************************************/

     LC_PO   = log(Chi_PO);            dC_PO  = dif(LC_PO);
     LEc_PO  = log(Ecu_PO);            dEc_PO = dif(LEc_PO);
     .                                 .
     .                                 .
     .                                 .
     LO_PO   = log(Other_PO);          dO_PO  = dif(LO_PO);

     WdC_PO  = Wbar1 * dC_PO;
     WdEc_PO = Wbar2 * dEc_PO;
     .
     .
     .
     WdO_PO  = Wbar8 * dO_PO;

     LL1_PPP = log(L1_PPP);            dL1_PPP = dif(LL1_PPP);
     LL2_PPP = log(L2_PPP);            dL2_PPP = dif(LL2_PPP);
     LL3_PPP = log(L3_PPP);            dL3_PPP = dif(LL3_PPP);

     WdL1_PPP = Wbar9  * dL1_PPP;
     WdL2_PPP = Wbar10 * dL2_PPP;
     WdL3_PPP = Wbar11 * dL3_PPP;
```

As indicated, mixed Divisia Index is also one of the mixed Rotterdam demand variables and is calculated in SAS as follows:

```
/*********************************************************************
*                    Mixed Divisia Index Calculation                 *
*                DQ = Σ_i w_ai Dq_ai + Σ_r w_br Dp_br                *
*********************************************************************/

     DQ =  WdC_PO + WdEc_PO + WdIi_PO + WdIo_PO + WdMx_PO
     + WdT_PO + WdV_PO + WdO_PO + WdL1_PPP + WdL2_PPP + WdL3_PPP;
run;
```

$$DQ = \sum_i w_{ai} Dq_{ai} + \sum_r w_{br} Dp_{br}$$

MRDS Parameter Estimation

- Model Procedure

The SAS PROC MODEL statement is used to estimate model parameters and elasticities. The PROC MODEL has a variety of features (SAS Institute Inc., 2013) useful for the estimation of a differential demand system as well as mixed Rotterdam demand model. Some of these features include:

- ✓ PROC MODEL statements; MODEL, FIT, SOLVE, TEST, OUTVARS, VAR, PARAMETERS, RETAIN, AND INSTRUMENTS.
- ✓ PROC MODEL options which can be applied with statements;
- o Options to control the estimation method; OLS, 2SLS, 3SLS, IT2SLS, IT3SLS, SUR, ITSUR, etc. General TEST statement options; WALD, LM, ALL, LR, OUT=,
- o Solution mode options and Monte Carlo simulation options.
- ✓ These statements and options allow flexibility in choosing the best method of estimation and inference.

- Adding Restrictions in Demand Systems.

The standard restrictions in estimating demand systems include adding up, homogeneity, and symmetry. There are two different approaches to impose these properties; 1) applying the RESTRICT statement with Model procedure, and 2) imposing restrictions directly in equations of system. For a mixed system the best approach is imposing these constrains directly in the model. In some other differential equations models such as the Almost Ideal Demand System (AIDS), this command works well, but it is not suggested for the mixed demand system. The adding up restriction in the data, which causes a singularity problem for the contemporaneous error covariance matrix, requires eliminating one equation from the model. In this paper the last equation which is the demand equation for the Gulf small size shrimp, L3, is randomly excluded prior to estimation. The estimates of the parameters are invariant to the deleted equation (Berndt and Savin, 1975). The Gulf small size shrimp demand parameters can be obtained by applying an adding-up restriction or re-estimating the system by dropping a different equation, randomly. In this instance, instead of running eleven equations, in a system, ten equations are included in MODEL procedure.

The homogeneity restriction is also imposed directly in the equations using the following identity.

$$\sum_i \theta_i - \sum_r \gamma_r = 1,$$

which is coded in SAS as follows:

c1 + c2 + c3 + c4 + c5 + c6 + c7 + c8 – s1 – s2 – s3 = 1,

To impose homogeneity restriction the equivalent of s3, which is the coefficient of small size shrimp, is included in the equations.

s3 = c1 + c2 + c3 + c4 + c5 + c6 + c7 + c8 – s1 – s2 – 1,

The PROC MODEL statement requires three phases;

Phase 1. Apply the “Data = option” in the PROC MODEL statement:
This option specifies the input SAS data set which contains the observed values of model variables (SAS/ETS User’s Guide, 2013).

Phase 2. Write equations of the system with the SAS programming statements.
This part contains the econometrics model written by SAS assignment statements, but not includes the error terms. The left-hand sides of the assignments are dependent variables and right-hand sides comprise independent/exogenous variables and coefficients of independent variables which should be estimated.

Phase 3. Use the FIT statement:
This statement fits equations of the system to the input data set to estimate the model parameters. The following SAS programming implements the above three phases.

Phase 1.

```
/*********************************************************************
*       Full Nonlinear Model for Mixed Rotterdam Demand System        *
*********************************************************************/
proc model data=mix6;
```

Phase 2.

```
/*********************************************************************
*        Share Equations for Mixed Rotterdam Demand System            *
*********************************************************************/

     WdC_PO = c1 * DQ + dc1*ld1 + dc2*ld2 + dc3*ld3 + t1*t
      + cc*c1* (dC_PPP - c1*dC_PPP - c2*dEc_PPP - c3*dIi_PPP
      - c4*dIo_PPP - c5*dMx_PPP - c6*dT_PPP - c7*dV_PPP - c8*dO_PPP
      + s1*dL1_PO + s2*dL2_PO + (c1+c2+c3+c4+c5+c6+c7+c8-s1-s2-1)*dL3_PO);
     .
     .
     .
     WdL2_PPP = - s2 * DQ + dl21*ld1 + dl22*ld2 + dl23*ld3 + t10*t
      - cc*s2* (dL2_PO - c1*dC_PPP - c2*dEc_PPP - c3*dIi_PPP
      - c4*dIo_PPP - c5*dMx_PPP - c6*dT_PPP - c7*dV_PPP - c8*dO_PPP
      + s1*dL1_PO + s2*dL2_PO + (c1+c2+c3+c4+c5+c6+c7+c8-s1-s2-1)*dL3_PO);
```

Phase 3.

```
   fit WdC_PO WdEc_PO WdIi_PO WdIo_PO WdMx_PO WdT_PO WdV_PO WdO_PO
   WdL1_PPP WdL2_PPP / itsur nestit dw=2 outest=est1 outs=s out=result1
   outall  converge= .0001  maxit = 1000;

parms
           cc c1 c2 . . . s2
              t1 t2 . . . t10
              dc1 dc2 . . . dL23;
     s3 = c1 + c2 + c3 + c4 + c5 + c6 + c7 + c8 -s1 -s2 -1;
run;
```

- The FIT Statement and OPTIONS

FIT statement includes dependent variables, the method of computing the system parameters, and some other options. One of the most important issues about the mixed Rotterdam demand system is the choice of the method to minimize the objective function and estimate model parameters. The PROC MODEL covers several methods such as Two-Stage Least Square (2SLS), Three-Stage Least Square (3SLS), Full Information Maximum Likelihood (FIML), Seemingly Unrelated Regression (SUR), and General Method of Moments (GMM). The best approach for minimization of a mixed Rotterdam demand system objective function is SUR which is a Feasible Generalized Least Square (FGLS). The reason for choosing this approach is that a mixed Rotterdam demand system consists of equations in which the error terms across equations are assumed to be correlated. The SUR method is also preferred to a maximum likelihood estimation method because this technique does not need any presumption about the distribution of equations' error terms. In applying the maximum likelihood estimation approach one should assume that the distribution of error terms, in a system, have multivariate normal distributions (Pindyck and Rubinfeld, 2008).

In this method, it is assumed that the system error terms are independent across time, but not across equations. That is, cross-equation contemporaneous correlations between error terms of the equations of the system exist. The SUR technique contains two steps of estimation. In the first step the OLS technique is implemented to achieve the residuals. Then the obtained residuals are comprised to estimate the elements of covariance matrix $\sum$. In second step the parameters of the system are estimated using the estimated covariance matrix, $\hat{\sum}$. The MODEL procedure also offers an iterated SUR (ITSUR). The ITSUR recalculates the matrix $\hat{\sum}$, after estimating the residuals from second step in first round. Then re-estimate the model parameters. This process will iteratively be continued until convergence is completed. In the FIT statement this is accomplish by adding the "ITSUR" option after slash.

The term NESTIT is also specified in the FIT statement. This option changes the default iteration method in SAS. This option allows for the inner parameter iterations for the fixed objective function. The "OUTS=" option saves the covariance matrix of the residuals across equations for later use. This matrix is known as an S matrix in SAS. This output is necessary for the estimation of the mixed Rotterdam demand elasticities. It can be recalled on a subsequent FIT or SOLVE statement. The option "OUTSET=" names a data set in which the estimated parameters are saved.

The "OUT=" option names a data set in SAS that includes estimated residuals by default. This data set can also contain predicted values and/or actual values in addition to the residuals, by selecting “OUTALL.” This option can be appropriated if “OUT=” option has been stated in the FIT statement.

When an iterative method is selected, one should specify the termination values for convergence and the maximum number of iterations. Options “CONVERGE=” and “MAXIT=” in FIT statement identify the end values for convergence and the maximum number of iterations, respectively. There are several convergence criteria. In PROC MODEL five convergence criteria are included and labeled as R, S, PPC, RPC, and OBJECT. For the purpose of estimation of mixed Rotterdam demand system parameters, the end value for convergence and the maximum number of iterations are equaled to 0.0001 and 1000, respectively.

Before running the model and obtaining the parameter estimates, the model parameters should be introduced. This is the task of the “PARMS” option. The last statement is the “RUN” command. The parameters of the mixed Rotterdam demand system can now be estimated. The MODEL procedure estimates the values of model’s parameters and prints results according to the chosen FIT statement options. The SAS output includes, the Mean Procedure, Model Summary, ITSUR Estimation Summary, Nonlinear ITSUR Summary of Residual Errors, and Nonlinear ITSUR Parameter Estimates. Some of these results are included in appendix.

- The Auto-Correlation and Seasonality Effects Tests

Prior to model estimation, the existence of serial correlation and seasonality should be checked. Shrimp imports and landings data displayed seasonality patterns, so that three quarterly seasonal dummy variables are included in every equation of the system. The last quarter dummy variable has been dropped to avoid the dummy variable trap, the full correlation matrix of variables. To detect the presence of autocorrelation in the residuals from the regression analysis the t-test are applied. For the t statistic test, each equation of the model is first estimated by OLS (equation by equation); next, OLS residuals, obtained from the first step, are regressed on all exogenous variables and the first and second lagged residuals (u_{t-1} and u_{t-2}). The parameters of these equations (ρ_1 and ρ_2) are the coefficients of the autoregressive equations (13). Therefore, the t statistics of these parameters (ρ_1 and ρ_2) are equivalent to the t statistics for the coefficient estimates of the autoregressive equations.

$$u_t = \rho_1 u_{t-1} + \rho_2 u_{t-2} + e_t \tag{13a}$$

$$\hat{u}_t = \hat{\rho}_1 u_{t-1} + \hat{\rho}_2 u_{t-2} . \tag{13b}$$

The advantage of this technique is that it allows one to detect the serial correlation problem even in models that use non-strictly exogenous explanatory variables (Wooldrige, 2008). A Durbin Watson (DW) test statistic is also performed. The Durbin Watson test statistic for AR(1) is also a valid test for this model, since the DQ is considered as constant (Theil, 1987) in the Rotterdam specification. A Durbin Watson test can easily be specified in MODEL procedure. One can add "DW=" option in FIT statement to obtain a Durbin Watson test results in SAS output.

For each equation of the mixed Rotterdam demand system, a null hypothesis of no serial correlation is constructed against the alternative of a first order autocorrelation process. Both t-test and DW test revealed that in some demand equations a first order autoregressive exist. To solve this problem the first order autocorrelation correction, AR(1) are considered in the model for all demand equations in the same manner. The AR(1) correction are specified in PROC MODEL right after demand equations and before the FIT statement.

```
%ar(WdC_PO, 1);
%ar(WdE_PO, 1);
.
.
.
%ar(WdL2_PPP, 1);
```

The second round Durbin Watson test illustrated an acceptable result, which means that autocorrelation is not a serious problem in the system.

In addition to AR(1) corrections, trend variables are added to the equations. The autoregressive correlation may have been overestimated[2] if the data demonstrates an upward or downward trend (Wooldrige, 2008). An upward time trend has been identified in imports observations, a common tendency in time series data. These variables are shown in the equations written in PROC MODEL statement (phase 2).

Performing a Monte Carlo Simulation

To evaluate the performance of the model, after imposing AR(1) and adding trend variables, a Monte Carlo simulation was designed and implemented using the SOLVE statement

[2] This issue refers to deterministic trends in the data, not stochastic trends. Author investigated whether the variables of the model which is studying here have the unit roots or stochastic trends. The unit root tests rejected the existence of stochastic trends.

following the FIT statement. Before specifying the SOLVE statement some issues should be considered. The MONTE CARLO simulation applies not only the estimated parameters to forecast but also the model's covariance matrix (Pindyck and Rubinfeld, 2008). The parameter covariance matrix provides random perturbations of the parameters for the forecasting process. The random perturbations have a multivariate normal distribution with expected value of 0 and the covariance matrix of the parameters and of the equation residuals. PROC MODEL is able to provide these two covariance matrices. The covariance matrix of the parameters can be provided by specifying "OUTEST=" and "OUTCOV" options in FIT statement. The covariance matrix of the equation residuals which is the cross-equation covariance matrix can also be contributed by stating "OUTS=" option in FIT statement. The MONTE CARLO simulation is performed by the SOLVE statement. The following SOLVE statement in the MODEL procedure performs a Monte Carlo simulation.

```
Solve WdC_PO WdE_PO WdIi_PO WdIo_PO WdM_PO WdT_PO WdV_PO WdO_PO
      WdL1_PPP WdL2_PPP / data=mix55 estdata=est1 sdata=s
      random=2000 seed=123 out=monte STATS THEIL;

       id year quarters;
       range year=2000;
run;
quit;
```

The SOLVE statement requires the input of dependent variables and the following options (after the "/"):

- Observations/variables data set; "DATA=",
- The covariance matrix of the parameter estimates data set; "ESTDATA=",
- The covariance matrix of the equation residuals which is the cross-equation covariance matrix data set; "SDATA=".

To perform Monte Carlo simulations the "RANDOM= " option is specified. The purpose of applying Monte Carlo simulations is to build up confidence intervals for error terms created by systematic random errors and by the estimated values of the parameters. Monte Carlo solution will be repeated as many number as given to the "RANDOM=" option for each BY group.

The “ID” statement sorts the observations and “RANGE” statement identified the solution year. In this study, observations are grouped by year and quarters and Monte Carlo solution will starts from 2000 to 2011, which is the last year of observations in data set.

The “SEED= ” option controls the random number generator for the simulations. The “SEED= ” option provides the same results if one repeat the performance of Monte Carlo simulations.

Following the MODEL procedure which includes FIT and SOLVE statements, a plot of the confidence intervals can be generated applying PROC UNIVARITE and PORC GPLOT statements for each dependent variable, separately. If the observations are not sorted, a PROC SORT statement should be driven prior to producing Confidence Intervals. PROC UNVARIATE creates percentile bounds and PROC GPLOT plots the graphs. The following SAS statements can be used to create a confidence interval for one of the model dependent variables, WdC_PO.

```
proc sort data=monte;
      by year quarters;
run;

proc univariate data=monte noprint;
      by year month;
      var WdC_PO;
      output out=C_bounds mean=mean p5=p5 p95=p95;
run;

title "Monte Carlo Generated Confidence Intervals”;
proc gplot data=C_bounds;
      plot mean*year p5*year p95*year /overlay;
      symbol1 i=join value=triangle;
      symbol2 i=join value=square l=4;
      symbol3 i=join value=square l=4;
run;
```

Estimating Mixed Rotterdam Demand Model Elasticities

The Rotterdam specification has been defined under double-log functions, allowing direct estimation of demand elasticities. These elasticities are uncompensated elasticities (Brown and

Lee, 2006). While the dependent variables are the budget shares times log differences, these elasticities are calculated at the mean point of budget shares.

In this study the mixed Rotterdam demand system includes eleven equations. Every equation comprises eleven price and quantity explanatory variables and elasticity estimates; therefore, this demand model contains 143 elasticity estimates. This MRDS is estimated with eight regular demand equations and three inverse demand equations. Accordingly, some of these elasticity estimates reflect flexibility estimates, but in a mixed demand specification all of them are considered elasticities.

There are two different approaches to estimate demand elasticities in the MODEL procedure. One of these two approaches generates t-test statistics for elasticities along with estimates. This is the approach that has been applied in this study using the ESTIMATE statement. Please note that the SOLVE statement is not necessary when the ESTIMATE statement is specified. The codes for elasticities should be provided right after the FIT statement and before the RUN statement. As noted earlier, elastisities of a MRDS are calculated at the mean point of budget shares. Therefore the mean value of each variable is identified after the FIT statement in MODEL procedure. To calculate elasticities the model parameter estimates are also needed. These estimated are stored in an ESTDATA data set by identifying “OUTEST=” option in the FIT statement. It should be noted that the “OUTCOV” option should not be specified in the FIT statement. This option adds parameters’ covariance matrix to the ESTDATA data set, while for the estimation of elasticities only estimated parameters are needed. The following SAS code demonstrates a small part of SAS statements for this purpose. All other elasticity formulae have the same SAS statements.

```
%let mw1=0.046510; %let mw2=0.097581;  ... %let mw11=0.036148;

     title 'Uncompensated Price and Quantity Elasticities';
     estimate 'e11=O-P Elasticity for China' (cc*(c1-c1*c1))/&mw1;
     estimate 'e12=C-P Elasticity for Chi-Eco' (cc*(  -c1*c2))/&mw1;
     .
     .
     .
     estimate 'e18=C-P Elasticity for Chin-Other'(cc*(  -c1*c8))/&mw1;
```

Summary

Consumer behavioral theory is a basic microeconomics theory which has been the backbone of almost all demand analysis studies. In addition to Marshallian demand systems and inverse demand systems which are two polar models, there is a third class of demand specification known as a mixed demand system. The mixed demand concept was first introduced by Samuelson (1965). This specification, which regards demand relationships as a function of mixed groups of quantities and prices, provides a more flexible specification than estimating a simultaneous system of demand and supply equations.

This article, for the first time, demonstrates how SAS statements and functions can provide reliable estimation of a mixed Rotterdam demand system, which is special case of mixed demand system. The mixed demand methodology and SAS programming described in this article is self-contained and can be adapted to a number of applications which use systems of demand equations. In this presentation, a mixed Rotterdam demand model was estimated, including a Monte Carlo simulation and elasticity estimates using SAS statements and options.

References

Barten, A.P. (1969). Maximum likelihood estimation of a complete system and demand equations. *Eurpean Economic Review*, *1* (1), 7-73.

Barten , A.P. (1992). The estimation of mixed demand system. in R. Bewley and Tran Van Hoe, eds. Contributions to consumer demand and econometrics: Essay in Honour of Henri Theil, London: Macmillan, 31-57.

Berndt, E.R., & N.E. Savin. (1975). Estimation and hypothesis testing in singular equation systems with autoregressive disturbances. *Econometrica*, *43* (5), 937-957.

Brown, M.G., J.Y. Lee, & X. Gao. (1993). A general differential demand model for mixed quantity- and price-dependent equations. Available at: http://purl.umn.edu/52735.

Brown, M.G., & J.Y. Lee. (2006). A mixed Rotterdam demand system. In M. Holt and J.P. Chanvas, eds. Essay in Honor of Stanley R. Johnson. Berekley Electroinc Press, Atricle 14. Available at: http://www.bepress.com/siohnson/art14/.

Chavas, J.P. (1984). The theory of mixed demand functions. *European Economic Review*, *24* (3), 321-344.

Cunha-e-Sa, M.A., & M.M. Ducla-Soares. (1999). Specification test for mixed demand system with an emphasis on combining contingent valuation and revealed data. *Journal of Environment Economics and Management*, *38* (2), 215-233.

Cunha-e-Sa, M.A., M.M. Ducla-Soares, L. C. Nunes, & P. Polome (2004). Consistency in Mixed Demand Systems: Contingent Valuation and Travel Cost Data. *American Journal of Agricultural Economics*, *86* (2), 444-454.

DEATON, A. & J. MUELBAUER (1980). An Almost Ideal Demand System. *The American Economic Review*, *70*(3), 312 – 326.

Gao, X. M., Wailes, E. J., & Cramer, G. L. (1996). Partial rationing Chinese urban household food demadn analysis. *Journal of Comparative Economics*, *22* (1), 43-62.

Gary Wong, K. K., & Park, H. (2007). The use of conditional cost functions to generate estimable mixed demand systems. *American Journal of Agricultural Economics*, *89* (2), 273–286.

Heien, D. (1977). Price determination processes for agricultural sector models. *American Journal of Agricultural Economics*, *59* (1), 126-132.

Moschini, G. and A. Vissa. (1993). Flexible specification of mixed demand system. *American Journal of Agricultural Economics*,*75* (1), 1-9.

McLaren K. R., & Gary Wong, K. K. (2009). Effective global regularity and empirical modeling of direct, inverse, and mixed demand systems. *Canadian journal of economics*, *42* (2), 749-770.

Moschini, G. and P.L. Rizzi. (2006). Coherent specification of a mixed demand system: The stone-geary model.: In M. Holt and J.-P. Chavas, eds. Essay in Honor of Stanley R. Johnson. Berkeley Electronic Press, Article 15. Available at: http://www.bepress.com/siohnson/art15/.

Moschini, G. and P.L. Rizzi. (2007). Deriving a flexible mixed demand system:
The normalized quadratic model. *American Journal of Agricultural ESconomics*, *89* (4), 1034–1045.

Pindyck, R. S. & Rubinfeld, D. L. (2008). *Econometric Models & Economic Forecasts* (4th ed.). New York City: McGraw-Hill publishing company.

Samuelson, P.A. (1965). Using full duality to show that simultaneously additive direct and indirect utilities implies unitary price elasticity of demand. *Econometrica, 33* (4), 781-796.

SAS Institute Inc. 2013. SAS/ETS® 12.3 User's Guide. Cary, NC: SAS Institute Inc.

Theil, H. (1975). *The theory and measurement of consumer demand: vol. 1*. Amsterdam: North-Holand.

Theil, H. (1976). *The Theory and Measurment of Consumer Demand: vol. 2.* Amsterdam: North-Holand.

Theil, H. & Clements, K. W. (1987). *Applied Demand Analysis: vol.7.* Cambridge, Massachusetts.